// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/SolidlyV2/Escrow/VeSolidEscrow.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity 0.8.11;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";

interface IVeV2 {
    function locked(uint256) external view returns (int128 amount, uint256 end);

    function merge(uint256 _from, uint256 _to) external;

    function safeTransferFrom(
        address _from,
        address _to,
        uint256 _tokenId
    ) external;

    function split(uint256 _from, uint256 _amount) external returns (uint256);

    function ownerOf(uint256 _tokenId) external view returns (address);

    function setApprovalForAll(address operator, bool _approved) external;

    function approve(address to, uint256 tokenId) external;

    function getApproved(uint256 _tokenId) external view returns (address);
}

interface IVeDistV2 {
    function claim(uint256 _tokenId) external returns (uint256);

    function claimable(uint256 _tokenId) external view returns (uint256);
}

interface IMinterV2 {
    function update_period() external returns (uint256);

    function active_period() external view returns (uint256);
}

interface IVoterV2 {
    function reset(uint256 _tokenId) external;
}

contract VeSolidEscrowManager is Ownable {
    using SafeERC20 for IERC20;

    struct EscrowData {
        bool goodStanding;
        uint128 tokenId;
        uint40 unlockTime;
    }

    address[] public escrows;
    mapping(uint256 => bool) public isEscrowed; // tokenId => is escrowed
    mapping(uint256 => address) public tokenIdToEscrow; // tokenId => escrow address
    mapping(address => EscrowData) public escrowData; // escrow address => tokenId

    mapping(uint256 => uint256) internal tokenIdLockedAmount; // tokenId => last locked amount

    // Solidly Addresses
    IVeV2 public immutable ve;
    IVeDistV2 public immutable veDist;
    IMinterV2 public immutable minter;
    IVoterV2 public immutable voter;

    mapping(bytes4 => bool) internal isBlockedVeMethod;

    // Temp storage just for creating escrows
    address public tempAddress;

    /**************************************** 
	 	 			 Events
	 ****************************************/
    event EscrowCreated(
        uint256 indexed tokenId,
        address indexed operator,
        address indexed escrow,
        uint256 unlockTime
    );

    event VeNftRevoked(uint256 indexed tokenId, address escrow);

    event Recovered(address tokenAddress, uint256 amountOrTokenId);

    event Standing(address escrowAddress, bool goodStanding);

    /**************************************** 
	 	 			Modifiers
	 ****************************************/

    /**
     * @notice Checks whether the veNFT stays in the escrow after a user interacts with the escrow
     */
    modifier veNftUnchanged(
        address escrow,
        address to,
        bytes calldata data
    ) {
        EscrowData memory _escrowData = escrowData[escrow];
        uint256 _tokenId = _escrowData.tokenId;
        require(_tokenId != 0, "Not an escrow");
        require(_escrowData.goodStanding, "Escrow in bad standing");

        bool _escrowInEffect = _escrowData.unlockTime > block.timestamp;

        if (_escrowInEffect) {
            // Check if method is approved if interacting with ve
            if (to == address(ve)) {
                bytes4 selector = bytes4(data);
                require(!isBlockedVeMethod[selector], "Cannot approve veNFTs");
            }

            // Update period if epoch changed
            if (block.timestamp >= minter.active_period() + 1 weeks) {
                minter.update_period();
            }
            tokenIdLockedAmount[_tokenId] = lockedAndClaimable(_tokenId);
        }

        _;

        if (_escrowInEffect) {
            // Check if veNFT is still in escrow
            require(ve.ownerOf(_tokenId) == escrow, "veNFT not in escrow");

            // Check if locked amount decreased
            require(
                lockedAndClaimable(_tokenId) >= tokenIdLockedAmount[_tokenId],
                "Locked amount decreased"
            );

            // Check if manger still has approval
            require(
                ve.getApproved(_tokenId) == address(this),
                "Manager no longer approved"
            );
        }
    }

    /**************************************** 
	 	 			Constructor
	 ****************************************/
    constructor(
        IVeV2 _ve,
        IVeDistV2 _veDist,
        IMinterV2 _minter,
        IVoterV2 _voter
    ) Ownable() {
        ve = _ve;
        veDist = _veDist;
        minter = _minter;
        voter = _voter;

        // Block escrows from approving veNFTs
        isBlockedVeMethod[IVeV2.approve.selector] = true;
        isBlockedVeMethod[IVeV2.setApprovalForAll.selector] = true;
    }

    /**************************************** 
	 	 		Restricted Methods
	 ****************************************/

    function createNewEscrow(
        uint256 tokenId,
        address operator,
        uint256 duration
    ) external onlyOwner {
        require(!isEscrowed[tokenId], "Already in escrow");
        isEscrowed[tokenId] = true;

        // Create escrow contract
        bytes32 salt = keccak256(abi.encode(tokenId, operator));

        tempAddress = operator;

        address escrowAddress = address(new VeSolidEscrow{salt: salt}());

        // Send veNFT to escrow
        ve.safeTransferFrom(msg.sender, escrowAddress, tokenId);

        // Approve manager to revoke veNFT if needed in the future
        (bool success, ) = VeSolidEscrow(escrowAddress)._executeCall(
            address(ve),
            0,
            abi.encodeWithSignature(
                "approve(address,uint256)",
                address(this),
                tokenId
            )
        );
        require(success, "Approval failed");

        // Record escrow address
        escrows.push(escrowAddress);
        EscrowData memory _escrowData = EscrowData({
            tokenId: uint128(tokenId),
            unlockTime: uint40(block.timestamp + duration), // Won't run into problems for 34000 years
            goodStanding: true
        });
        escrowData[escrowAddress] = _escrowData;
        tokenIdToEscrow[tokenId] = escrowAddress;

        emit EscrowCreated(
            tokenId,
            operator,
            escrowAddress,
            block.timestamp + duration
        );
    }

    /**
     * @notice Used to set an escrow's standing (to true = good or false = bad)
     */
    function setStanding(uint256 tokenId, bool standingStatus)
        external
        onlyOwner
    {
        address escrowAddress = tokenIdToEscrow[tokenId];
        require(escrowAddress != address(0), "Not an escrow");
        EscrowData memory _escrowData = escrowData[escrowAddress];
        require(_escrowData.unlockTime > block.timestamp, "Escrow expired");

        // Change and emit standing state if different
        if (_escrowData.goodStanding != standingStatus) {
            escrowData[escrowAddress].goodStanding = standingStatus;

            emit Standing(escrowAddress, standingStatus);
        }
    }

    /**
     * @notice Used to detach gauges before revoking
     */
    function detachGauges(uint256 tokenId, address[] calldata gauges)
        external
        onlyOwner
    {
        address _escrow = tokenIdToEscrow[tokenId];
        require(!escrowData[_escrow].goodStanding, "Escrow in good standing");

        bytes memory data = abi.encodeWithSignature(
            "withdrawToken(uint256,uint256)",
            0,
            tokenId
        );

        for (uint256 i = 0; i < gauges.length; i++) {
            VeSolidEscrow(_escrow)._executeCall(gauges[i], 0, data);
        }
    }

    /**
     * @notice Used to reset votes before revoking
     */
    function resetVotes(uint256 tokenId) external onlyOwner {
        address _escrow = tokenIdToEscrow[tokenId];
        require(!escrowData[_escrow].goodStanding, "Escrow in good standing");

        voter.reset(tokenId);
    }

    /**
     * @notice Revokes the veNFT if a user misbehaves
     */
    function revokeNft(uint256 tokenId) external onlyOwner {
        address _escrow = tokenIdToEscrow[tokenId];
        require(!escrowData[_escrow].goodStanding, "Escrow in good standing");

        // Transfer veNFT
        ve.safeTransferFrom(_escrow, owner(), tokenId);

        // Update states
        isEscrowed[tokenId] = false;
        EscrowData memory _escrowData = EscrowData({
            goodStanding: true, // So the user can still interact with the contract after the veNFT is revoked
            tokenId: uint128(tokenId),
            unlockTime: uint40(block.timestamp) // So the user can still interact with the contract after the veNFT is revoked
        });
        escrowData[_escrow] = _escrowData;

        emit VeNftRevoked(tokenId, _escrow);
    }

    function recoverERC20(address _tokenAddress, uint256 _tokenAmount)
        external
        onlyOwner
    {
        IERC20(_tokenAddress).safeTransfer(owner(), _tokenAmount);
        emit Recovered(_tokenAddress, _tokenAmount);
    }

    /**************************************** 
	 	 		 View Methods
	 ****************************************/

    function lockedAndClaimable(uint256 tokenId)
        internal
        view
        returns (uint256)
    {
        (int256 lockedAmount, ) = ve.locked(tokenId);
        uint256 claimableAmount = veDist.claimable(tokenId);
        return uint256(lockedAmount) + claimableAmount;
    }

    /**************************************** 
	 	 		 Wrapper Methods
	 ****************************************/
    /**
     * @notice Called by escrow contracts, checks whether interactions jeopardizes the veNFT
     */
    function wrappedExecuteCall(
        address to,
        uint256 value,
        bytes calldata data
    ) external payable veNftUnchanged(msg.sender, to, data) {
        (bool success, bytes memory returnData) = VeSolidEscrow(msg.sender)
            ._executeCall{value: msg.value}(to, value, data);

        require(success == true, "Transaction failed");
    }

    /**************************************** 
                     ERC721
     ****************************************/

    /**
     * @notice This contract should not receive NFTs
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4) {
        revert("This contract doesn't accept NFTs");
    }
}

/**
 * @notice  Manager should not be able to do anything other than revoking the veNFT
 * 			Operator should be able to do anything other than approvals as long as the
 * 			veNFT remains in the escrow
 */
contract VeSolidEscrow {
    address public immutable manager; // Immutable so there's no way to bypass onlyManager

    mapping(address => bool) public isOperator; // People who can access the veNFT in this contract

    /**************************************** 
                      Events
     ****************************************/

    event OperatorStatus(address indexed operator, bool state);

    /**************************************** 
                     Modifiers
     ****************************************/
    modifier onlyManager() {
        require(msg.sender == manager, "Only manager");
        _;
    }

    modifier onlyOperator() {
        require(isOperator[msg.sender], "Only Operator");
        _;
    }

    /**************************************** 
                    Constructor
     ****************************************/

    constructor() {
        manager = msg.sender;
        address _operator = VeSolidEscrowManager(msg.sender).tempAddress();
        isOperator[_operator] = true;
        emit OperatorStatus(_operator, true);
    }

    /**************************************** 
                 User Interactions
     ****************************************/
    /**
     * @notice Sets operator status
     * @dev Operators are also allowed to add other operators
     */
    function setOperator(address operator, bool state) external onlyOperator {
        if (isOperator[operator] != state) {
            isOperator[operator] = state;
            emit OperatorStatus(operator, state);
        }
    }

    /**
     * @notice Allows the user to do anything except approving the veNFT
     * @dev Manager checks whether the veNFT stays in escrow at the end of the tx
     */
    function executeCall(
        address to,
        uint256 value,
        bytes memory data
    ) external payable onlyOperator {
        VeSolidEscrowManager(manager).wrappedExecuteCall{value: msg.value}(
            to,
            value,
            data
        );
    }

    /**************************************** 
                  Wrapped Call
     ****************************************/

    /**
     * @notice Internal notation because this is only reachable via executeCall() callable by operators
     * @dev The only time manager's owner has access to this is during revoking which approves the
     * 		veNFT to the manager and detaches from gauges.
     */
    function _executeCall(
        address to,
        uint256 value,
        bytes memory data
    )
        external
        payable
        onlyManager
        returns (bool success, bytes memory returnData)
    {
        (success, returnData) = to.call{value: value}(data);

        // Bubble revert reason up if reverted
        if (!success) {
            assembly {
                returndatacopy(0, 0, returndatasize())
                revert(0, returndatasize())
            }
        }
    }

    /**************************************** 
                     ERC721
     ****************************************/

    /**
     * @notice Mandatory ERC721 receiver
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4) {
        return this.onERC721Received.selector;
    }
}